
Introduction Requirements

I Introduction

OoRexx.NET is an approach for bridging the .NET framework and Open
Object Rexx. It was developed by student Manuel Ra�el of the University of
Economy Vienna. To accomplish that, BSF4ooRexx is used to get access to
Java classes. It then camou�ages Rexx classes as if they were .NET classes
so it is possible to use the entire framework with all of its advantages. Open
Object Rexx is an object-oriented programming language based on Rexx,
which was invented in 1979 by IBM.
This bachelor thesis is about exploring the many attributes, methods, classes,
namespaces and assemblies from the .NET framework and implement it in
the dynamically typed, caseless and easy-to-use ooRexx. It includes both
basic, easy to understand examples and more advanced ones which might
take a while for the non-experienced reader to understand. Although the
author tried to keep the application code well-commentated and therefore
well-documented, it is a good approach to �rst see the outcome of each ap-
plication before looking at the code. To achieve that the reader could use
the "viewAll.rxj" �le included in the directory. It launches all applications
one-by-one with the "rexxpaws.exe" program, which pauses after each ap-
plication, so that the reader can choose when to start the next one. After
exiting the last application, it will delete all temporary �les created in the
process to keep the directory clean.

1 Requirements

In order to use this Cookbook examples or create ooRexx.NET applications
on your own, the reader will need to have the following programs installed
on your computer:

• Open Object Rexx, current version

• BSF4ooRexx, current version

• Java, Version 1.5 or higher

• Microsoft .NET Framework, Version 4 or higher

• Jni4net

Be aware, that installing those components with di�erent bitnesses (32 or 64
bit) might eventually cause problems.

2 Step-by-step ooRexx.NET Installation Guide

This setup guide will help the reader to install all necessary components
of ooRexx.NET and its prerequisites. It can be assumed that the reader
already has access to the zipped ooRexx.NET �les. They will be distributed
alongside BSF4ooRexx version 4.52. in fall 2016. If the reader has some parts
of this installation guide already installed and ready to go on your computer,
the reader can simply skip them. It is recommended though to always use
the current versions due to security patches and function extensions - so
better have a look at the used version of already installed components before
skipping.

1. Download and install ooRexx. A list of all available versions can be
found via the following link: https://sourceforge.net/projects/

oorexx/files/oorexx/.

2. Download and install Java. Most systems already have Java installed
because it is so popular, but take care that the reader has at least
version 1.5 installed. the reader can get it from that the Oracle page
directly: https://www.java.com/de/download/.

3. Download and install the .NET framework from the Microsoft web-
site. Here is a link to the web installer: https://www.microsoft.com/
de-at/download/details.aspx?id=30653.

4. Download and install BSF4ooRexx. This is the link to the latest
version:
https://sourceforge.net/projects/bsf4oorexx/files/latest/.
Use the installer provided in the following path:
"install/windows/install.cmd".

5. Go into the �le settings of the ooRexx.NET zip archive. At the bot-
tom of the "general" tab the reader will �nd a button, which marks this
archive and all of its components safe and unblocks all of its contents.
See Figure 1 for a screenshot of that form. After doing so the reader can
unzip that zip archive and start install_clr_support.cmd. This exe-
cutable Windows Batch �le will add jni4net.n-0.8.8.0.dll and oorexx.net
.dll to the Global Assembly Cache (GAC) [1]. Both �les are needed for
ooRexx.NET to bridge the .NET framework using Java. DLL stands
for "dynamic-link library" and is a collection of .NET classes.

Figure 1: File settings of ooRexx.NET.zip. Activate this button at the bottom of this
form to mark the zip archive as safe, which allows the installer to work properly. In an
English Version of Windows, the button is labelled "Unblock".

6. CLR.CLS: This is an Open Object Rexx �le containing all ooRexx
.NET classes and its members, as well as several public routines. It
also requires BSF.CLS itself, making it possible to use both .NET and
Java classes with just one ::REQUIRES call. For this directive to �nd this
�le in your applications, simply copy CLR.CLS into the BSF4ooRexx
directory.

7. jni4net.j-0.8.8.0.jar: It is important to point the environmental vari-
able "CLASSPATH" to this JAR �le directly. To access this variable,
use Windows' Command Prompt with the following command:
"SET CLASSPATH=%CLASSPATH%;[absolute path to the jar �le];".
It will maintain the old value of the PATH variable and add a new
entry. Overriding some environmental variables can cause system sta-
bility issues. Sometimes a system reboot is needed in order to �nish the

variable changing process. If the environmental variables do not save
after closing the cmd.exe application, there might be a problem with
access control. In that case, open the System Control Panel and search
for environmental variables. Using that form, the reader can de�ne and
change both environmental variables for the current user and for the
system.

Figure 2: changing the environmental variables in Windows' System Control Panel

8. oorexx.net.jar: Finally, include this JAR �le into the classpath vari-
able as described in the previous step. [2, 3]

3 ooRexx.NET in a Nutshell

Here are the 5 most important things to know about ooRexx.NET program-
ming in a nutshell:

1. CLR.CLS is the only needed ooRexx �le to access the .NET frame-
work. Include it in all your applications with the ::REQUIRES directive
of ooRexx. Keep in mind, that if problems occur there is always the
possibility to have a look at the CLR.CLS code and maybe �nd a so-
lution there. It is well-structured and also well-documented with a lot
of comments to have a better understanding of the code.

2. Use .clr~new(CLASSNAME, argument1, argument2, ...) to instantiate
a class with constructors. If this class has one or more constructors,
it is possible to get an instance of it. The easiest and de�nitely most
popular constructor has no arguments, so that is how it looks like in
the .NET documentation: CLASSNAME().

3. Use clr.import(CLASSNAME) to import a static class, which has no con-
structors and therefore does not accept any input variables when in-
stantiating it. Additionally, this routine o�ers to import normal classes
as well and save it into an ooRexx class for future use, as this coding
example shows:

1 CALL clr.import "System.IO.StreamWriter", "netStreamWriter" -- import class, save it in

local as "netStreamWriter"

2

3 file = .netStreamWriter~new("tmp.txt") -- create a streamwriter object

4 file~write("hello, .net world!")

5 file~close

6

7 ::REQUIRES CLR.CLS

Listing 1: Using the clr.import routine to import a .NET class and use it as if it was
an ooRexx class

This is especially useful for classes which are used multiple times in an
application to make it more e�cient. The clr.import routine imports
the class stated in the �rst argument into the second argument. From
that point it can be used like a regular ooRexx class.

4. Treat enumerations like static classes. Treat enumeration values as if
they were attributes.

5. Use Microsoft's Developer Network (MSDN, https://msdn.microsoft.
com/en-us/library/) e�ectively to �nd what the reader is searching

for. Microsoft's programming languages include C#, F#, Visual Ba-
sic, ASP.NET and many more. They have built a great community and
are very likely to have already stumbled upon a problem we ooRexx
programmers can use to try and solve our own.

For further information on ooRexx.NET have a look at [4].

II Coding Examples

1 01-systemsounds.rxj

This application shows how to get access to the static .NET class "System-
Sounds" and how to use it properly.

1 /* File: 01-systemsounds.rxj

2 * Description: This application demonstrates all available SystemSounds

3 * Shows:

4 * - calling the Rexx Method "SysSleep" to do nothing for a specific amount

5 * of seconds

6 * - Access to static .NET classes

7 * - Console Output via .NET

8 * - Using SystemSounds class to play specific system sounds

9 * - Including ooRexx.NET with the REQUIRES directive

10 */

11

12 sounds = clr.import("System.Media.SystemSounds")

13 console = clr.import("System.Console")

14

15 console~WriteLine("SystemSounds demonstration starting") -- the method "WriteLine" is

basically identical to ooRexx' SAY command. If no line feed is needed after the

output, you should use "Write" instead of "WriteLine".

16 CALL SysSleep .5 -- wait for 500 ms

17

18 console~WriteLine("playing 'Beep'")

19 sounds~Beep~Play

20 CALL SysSleep 1

21

22 SAY "playing 'Asterisk'"

23 sounds~Asterisk~Play

24 CALL SysSleep 1

25

26 SAY "playing 'Exclamation'"

27 sounds~Exclamation~Play

28 CALL SysSleep 1

29

30 SAY "playing 'Hand'"

31 sounds~Hand~Play

32 CALL SysSleep 1

33

34 SAY "the last one is called 'Question'"

35 sounds~Question~Play

36 CALL SysSleep 1

37

38 ::REQUIRES CLR.CLS -- get ooRexx.NET support

Listing 2: Using SystemSounds and Console in ooRexx.NET

Figure 3: Output of 01-systemsounds.rxj

"System.Media.SystemSounds" is a static class and cannot be instanti-
ated. Line number 12 shows the public routine clr.import() from CLR.CLS
to import a static class by giving the fully quali�ed name of the class as
an argument. Beep, Asterisk, Exclamation, Hand and Question are static
properties of the class "SystemSounds". Accessing them will get the sound
associated to the corresponding program event in the current Windows sound
scheme. Note that some sound schemes do not have any sounds speci�ed for
those events, so it might be better to use other sound classes with �xed .mp3
sounds - or to use system sounds as an additional source of con�rmation for
user behaviour.

[5, 6]
The output of this application is very straight forward and can be seen

in Figure 3.

2 02-streamwriter.rxj

This application shows the usage of a �le writer in ooRexx.NET.

1 /* File: 02-streamwriter.rxj

2 * Description: This application creates a new textfile in the same directory and writes

some text into it

3 * Shows:

4 * - Usage of static System.IO.Directory class to get the path to the

5 present working directory using .NET support

6 * - Instantiating a .NET class

7 * - Creating a new textfile

8 * - Writing into that textfile

9 * - Getting the current encoding which can be changed if needed

10 * - Closing a filestream

11 */

12

13

14 directory = clr.import("System.IO.Directory") -- get access to static class

System.IO.Directory

15 filedirectory = directory~GetCurrentDirectory -- this method returns the path of the

current directory of this application. This assures, that this application works on

each and every system, regardless of where the files are saved.

16 -- alternatively, use rexx's built-in function directory() to get the current directory

17 filename = "02-textfile.txt"

18 filepath = filedirectory || .file~separator || filename -- .file~separator: on Windows

"\", on Unix "/"

19

20

21 file = .clr~new("System.IO.StreamWriter", filepath) -- get an instance of the public class

"System.IO.StreamWriter"

22

23 file~WriteLine("~~First Heading~~")

24 file~WriteLine

25 file~WriteLine("This ooRexx.NET application will create a new textfile in given file path

if it does not exist already.")

26

27 encoding = file~Encoding -- get access to enconding of the file, which will be most likely

UTF8

28 SAY "used encoding:" encoding~toString

29

30 file~Close -- close the opened file to free some memory

31 SAY "The textfile was successfully created."

32

33

34 ::REQUIRES CLR.CLS

Listing 3: Writing and creating �les in ooRexx.NET

Figure 4: Output of 02-streamwriter.rex

The public class "System.IO.StreamWriter" has many constructors like

an already open stream or the choice of an encoding, although in most cases
a �lepath will be su�cient. Lines 14 and 15 generate the absolute path of the
current directory of this application using the .NET class "System.IO.Direct-
ory". Alternatively, ooRexx provides the directory() routine. The output
of variable "�lepath" de�ned in Line 18 is dynamic and will look di�erently
on every system, depending on the location of the application. In my case,
this is how the variable looks like:
"C:\Users\user\Documents\ooRexx.NET\samples\02-text�le.txt"
Note that in Windows absolute paths are always separated with a backslash,
while other operating systems such as Linux use normal slashes.
The output of this program can be viewed in Figure 4. There the reader can
also see the default encoding "System.Text.UTF8Encoding", which contains
a large number of special characters and thus can represent almost anything.
There are many other encodings - while utf8 uses at least 1 byte for each
letter - utf16 uses twice or utf32 even 4 times the amount of disk space per
letter. If �le size matters and only english letters are used it is sometimes
better to stick to basic encodings like ASCII. [7]

3 03-streamreader.rxj

This application is based on the previous coding example from Listing 3 since
it uses the text�le, which was created with the stream writer.

1 /* File: 03-streamreader.rxj

2 * Description: This application opens a file and outputs it using different techniques

3 * Shows:

4 * - Using a linefeed and pretty print routine

5 * - Difference between ReadToEnd, ReadLine and Read

6 * - Using the class "System.Convert" to convert a number representing a char

into a real char

7 */

8

9 SIGNAL ON SYNTAX -- in case an error occurs, go to Syntax jumping point in Line 54

10 returnPointAfterSyntax: -- define a returning point

11

12 lf = "0a"x -- defining a linefeed

13 directory = clr.import("System.IO.Directory")

14 filedirectory = directory~GetCurrentDirectory

15 filename = "02-textfile.txt"

16 filepath = filedirectory || .file~separator || filename

17

18 /* get access to static classes System.Console and System.Convert for future use */

19 console = clr.import("System.Console")

20 converter = clr.import("System.Convert")

21

22 fileStream = .clr~new("System.IO.StreamReader", filepath) -- open a file stream with the

file created in 02-streamwriter.rxj

23

24 SAY "Output generated with [ReadToEnd]:" lf

25 prettyOutput = pp(fileStream~ReadToEnd) -- pp stands for pretty print and is a rexx method

to add square brackets before and after the given argument

26 console~Write(prettyOutput)

27 fileStream~Close -- close the filestream to free resources

28

29 SAY lf "------------------" lf

30

31

32 fileStream = .clr~new("System.IO.StreamReader", filepath) -- open the filestream again

33 SAY "Output generated with [ReadLine]:" lf

34

35 DO UNTIL fileStream~Peek = -1 -- -1 means, that there is no further line to output

36 prettyOutput = pp(fileStream~ReadLine)

37 console~WriteLine(prettyOutput) -- WriteLine is used to insert a line break afterwards,

which we need here because we read each line

38 END

39 fileStream~Close

40

41 SAY lf "------------------" lf

42

43

44 fileStream = .clr~new("System.IO.StreamReader", filepath) -- open the filestream again

45 SAY "Output generated with [Read]:" lf

46 DO WHILE fileStream~Peek <> -1 -- while there is a character available

47 char = converter~ToChar(fileStream~Read)

48 prettyOutput = char

49 console~Write(prettyOutput)

50 END

51 fileStream~Close

52 exit

53

54 Syntax: -- assuming the syntax error was caused because of the missing file

"02-textfile.txt"

55 "rexx ""02-streamwriter.rex""" -- call the streamwriter application to create

02-textfile.txt

56 SIGNAL returnPointAfterSyntax -- return to Line 10

57

58 ::REQUIRES CLR.CLS -- get ooRexx.NET support

Listing 4: Reading �les in ooRexx.NET

Figure 5: Output of 04-streamreader.rxj

A few commands are the same as in the last example, like getting the
present working directory with the public method
"Directory.GetCurrentDirectory". [8]
It starts by de�ning a linefeed variable, which can also be called "End Of
Line" character or "Line Break", which is exactly what it does - a line break.
Since every Rexx SAY commands ends with a line break, it is also possible
to use multiple SAY commands instead of linefeeds, but those keep the code
clean and easy to read. The "0a" from lf = "0a"x uses hexadecimal encod-
ing, which is equivalent to decimal spelling 10 - and "x" implies that we want
to access a value from an ASCII table, in that case the 10th value, which is
a line break.
fileStream~peek returns an integer representing the next character to be
read or -1 in case there are no characters left to read, signalising that the
document has reached its end. The focus of this application is to show the
di�erences between "ReadToEnd", "ReadLine" and "Read". ReadToEnd returns
a string containing all of the characters in the text�le, including special char-
acters like tabulators or carriage returns. ReadLine reads until the next line

break and sets the cursor afterwards, so that the next call to ReadLine or
Read will start at that point. In contrast to ReadLine, Read will receive only
the next character as an integer as opposed to the whole line.

4 04-messagebox.rxj

This is a very easy and basic coding example which shows how to access
.NET message boxes.

1 /* File: 04-messagebox.rxj

2 * Description: Shows a Message Box in ooRexx.NET with a title and a text

3 * Shows: - MessageBox.Show method with title and a text

4 */

5

6 text = "This is my Text" -- define text

7 title = "Title of MessageBox" -- define title

8 MessageBox = clr.import("System.Windows.Forms.MessageBox") -- get access to static class

"MessageBox" of namespace "System.Windows.Forms"

9 MessageBox~Show(text, title) -- start method "show" with two arguments: text and optional

title

10

11 ::REQUIRES CLR.CLS -- get ooRexx.NET Support

Listing 5: Message Boxes in ooRexx.NET

A message box is a nice way to communicate with the user or to display
error/success messages. It is used in many programming languages with dif-
ferent names, like Message Dialog in Java or Message Box in Python. How
it's displayed is dependent on the operating system and .NET framework ver-
sion. It is not possible to change either the width or height of a message box
- making it an unreliable but still easy-to-implement part of user interaction.

Figure 6: The default MessageBox in .NET

5 05-messagebox.advanced.rxj

This coding example shows just how complex message boxes can get. Some-
times it is useful to give the user a choice when clicking on a message box.
This is where additional buttons come in handy, which can be passed as the
third argument in the MessageBox~Show() method.

1 /* File: 05-messagebox.advanced.rxj

2 * Description: Shows multiple Message Boxes with different button setups

3 * Shows: - MessageBox with Yes, No and Cancel options

4 * - MessageBox with OK and Cancel options

5 * - MessageBox with Abort, Retry and Ignore options

6 * - Reacting on user interaction

7 */

8

9 title = "Title of MessageBox" -- define title

10 -- get access to the static class "MessageBox" of namespace "System.Windows.Forms", a

"Dialog" window

11 MessageBox = clr.import("System.Windows.Forms.MessageBox")

12 DialogResult = clr.import("System.Windows.Forms.DialogResult") -- get access to

DialogResult enumerations"

13

14 -- get access to MessageBoxButtons enumerations

15 MessageBoxButtons = clr.import("System.Windows.Forms.MessageBoxButtons")

16

17 -- display first message box and save result in Rexx variable "res"

18 title = 'Message Box # 1'

19 res = MessageBox~show('This is a YesNoCancel Messagebox. Press the "Yes" button to

continue', title, MessageBoxButtons~YesNoCancel)

20

21 IF res = DialogResult~Yes THEN DO -- compare message box result with the enumeration

value defined in DialogResult

22 SAY title 'returned "OK"'

23 DO LABEL outer FOREVER -- define a DO block with a name (as a possible target

for LEAVE in the inner DO block)

24 title='Message Box # 2 ("outer")'

25 res = MessageBox~show('This is another Messagebox (you pressed the "'res~toString'"

button). Press the "OK" button to see the next one', title,

MessageBoxButtons~OKCancel)

26 IF res = DialogResult~OK THEN -- compare message box result with the enumeration

value defined in DialogResult

27 DO

28 SAY title 'returned "OK"'

29 DO LABEL inner FOREVER

30 title = 'Message Box # 3 ("inner")'

31 res = MessageBox~show('Messagebox # 3 (you pressed the "'res~toString'" button).

Click whatever you want.' , title, MessageBoxButtons~AbortRetryIgnore)

32 IF res = "Abort" THEN -- compare with a string

33 DO

34 SAY title 'returned "Abort"'

35 LEAVE outer -- leave do-block named "outer"

36 END

37 IF res = "Retry" THEN -- leave do-block named "inner"

38 DO

39 SAY title 'returned "Retry"'

40 LEAVE inner -- leave do-block named "inner"

41 END

42 SAY title 'returned "Ignore"'

43 END

44 END

45 ELSE

46 DO

47 SAY title 'returned "Cancel"'

48 LEAVE outer

49 END

50 END

51 END

52 ELSE

53 SAY title 'returned "'res~toString'"'

54

55 ::REQUIRES CLR.CLS -- get ooRexx.NET support

Listing 6: Advanced Message Boxes in ooRexx.NET

Take a second look at Line 15 where we de�ne the Rexx variable Mes-
sageBoxButtons as "System.Windows.Forms.MessageBoxButtons" class. This
is actually a public enumeration - which is a set of prede�ned constants which
can only be used for one speci�c reason, in this particular case for message
boxes. This construct can be seen quite often in the .NET framework, that's
why this example shows how to handle them correctly.
Line 14 de�nes another set of enumerations as the Rexx variable "DialogRe-
sult" to indicate the return value of a message box. The members are:

• OK

• Yes

• No

• Abort

• Cancel

• Retry

• Ignore

• and �nally None if nothing was returned from the message box.

Line 21 compares the result of the shown message box against a DialogRe-
sult enumeration, in this example "Yes". The next button presses compare
directly to the Rexx strings "Abort", "Retry" and "Ignore", which is also
possible. [9, 10]

Figure 7 shows the �rst message box providing the Yes/No/Abort options.

Figure 7: A Yes/No/Abort MessageBox in ooRexx.NET

The second message box shows when clicking on "Yes" in the previous
one.

Figure 8: An OK/Abort MessageBox in ooRexx.NET

The third message box provides the possibility to show the second message
box again when activating the "Retry" option.

Figure 9: An Abort/Retry/Ignore MessageBox in ooRexx.NET

Meanwhile, all user input gets documented in the terminal in Figure 9.

Figure 10: The terminal output of 05-messagebox.advanced.rxj

6 06-process.demonstration.rxj

This example demonstrates process handling in ooRexx.NET. It opens a text
�le in notepad.exe, a website in a web browser and sends the F11 key to toggle
fullscreen mode in Internet Explorer.

1 /* File: 04-process.demonstration.rxj

2 * Description: This application demonstrates object-oriented process handling in

ooRexx.NET

3 * Shows:

4 * - Usage of basic message boxes with a specified title

5 * - Starting a process by dispatching "Start"

6 * - Sending keys to manipulate running applications

7 * - Starting a process with maximized window style and additional

information

8 * - Retrieving the absolute path to temporary folder

9 */

10

11

12

13 directory = clr.import("System.IO.Directory") -- get access to static class

System.IO.Directory

14 filedirectory = directory~GetCurrentDirectory -- get path to present working directory

15 filename = "02-textfile.txt"

16 filepath = filedirectory || .file~separator || filename -- absolute path to the textfile

from sample 02-streamwriter.rex

17 myProcess = .Process~new -- get a reference to class "Process" of this file (Line 67)

18 messageBox = clr.import("System.Windows.Forms.MessageBox") -- preload messageBox class

into a variable for further use

19

20 messageBox~show("Let's open our new textfile", "Information") -- open a messageBox with

the first argument as text and second argument as title

21 myProcess~Application = "notepad" -- define class attribute "Application" -> set it to

"notepad". No file extensions needed - the .NET framework knows what we want

22 myProcess~Filepath = filepath -- define class attribute "Filepath" and set it to our

textfile

23 myProcess~Start -- start process "notepad" and open the file given in variable filepath

24 CALL SysSleep 2 -- wait for 2 seconds

25

26 messageBox~show("Now I will open a Website in your favorite Webbrowser", "Information")

27 myProcess~Reset -- call to class method "Reset" which resets all the values so another

Process can be started with the same class instance

28 myProcess~Filepath = "https://google.com"

29 myProcess~Start(.true) -- open this website with the standard application for the "https"

protocol, e.g. standard web browser

30 CALL SysSleep 2

31

32 messageBox~show("Of course I can force you to use Internet Explorer", "Information")

33 myProcess~Application = "IExplore"

34 myProcess~Filepath = "https://google.com" -- start Internet Explorer and open the link

specified in Filepath

35 myProcess~Start

36 CALL SysSleep 2

37

38 messageBox~show("I can even toggle Fullscreen by emulating keyboard presses",

"Information")

39 sendKeys = clr.import("System.Windows.Forms.SendKeys") -- get access to static class

System.Windows.Forms.SendKeys

40 sendKeys~SendWait("{F11}") -- this is a call to the method "SendWait" which presses F11

and waits for a response. F11 toggles fullscreen mode in Internet Explorer.

41 CALL SysSleep 1

42 sendKeys~SendWait("{F11}") -- return to normal mode

43 CALL SysSleep 2

44

45

46 path = clr.import("System.IO.Path") -- get reference to static class "System.IO.Path"

47 tempPath = path~GetTempPath -- --Path.GetTempPath returns a string with the absolute path

to the temporary folder which usually looks like this but can vary for each system:

C:\Users\[UserName]\AppData\Local\Temp\. Alternatively, it can be accessed using

%TEMP%

48 tmpfilepath = tempPath || .file~separator || "temporary.txt"

49 file = .clr~new("System.IO.StreamWriter", tmpfilepath) -- open a file stream

50 file~Write("It's also possible to specify additional information, like WindowStyle to

maximize or minimize the window.")

51 file~Write("(Program has ended, leaving this window open.)")

52 file~Close

53

54 ProcessStartInfo = .clr~new("System.Diagnostics.ProcessStartInfo", "Notepad") --

ProcessStartInfo is used to set a few values before starting a process

55 ProcessStartInfo~Arguments = tmpfilepath -- give notepad the absolute file path to our new

textfile as an argument

56 ProcessWindowStyle = clr.import("System.Diagnostics.ProcessWindowStyle")

57 ProcessStartInfo~WindowStyle = ProcessWindowStyle~Maximized -- now the application

"Notepad" will start with window style maximized.

58

59 myNewProcess = .Process~new

60 myNewProcess~ProcessStartInfo = ProcessStartInfo

61 myNewProcess~Start -- start a new process based on the information provided in

ProcessStartInfo

62

63

64 ::REQUIRES PROCESS.CLS -- get access to the Process class which includes CLR.CLS

(ooRexx.Net support)

Listing 7: Process handling in ooRexx.NET

The PROCESS class is de�ned in PROCESS.CLS. This �le requires
CLR.CLS and provides multiple methods for easier process handling.

1 /* File: PROCESS.CLS

2 * Class: Process

3 * Description: Starts Windows Processes based on information about application or file

4 */

5

6

7 ::CLASS Process PUBLIC

8 ::ATTRIBUTE Application

9 ::ATTRIBUTE Filepath

10 ::ATTRIBUTE ProcessStartInfo

11

12 /* Method: init

13 * Description: Constructor which saves the passed arguments in the class and defines

variable "Process" as .NET class "System.Diagnostics.Process"

14 * Argument: - Application: the application which will be started

15 */

16 ::METHOD init

17 EXPOSE Application Process

18

19 if arg(1,"Exists") then -- if an argument was supplied, assign it to attribute

"application"

20 application = arg(1)

21

22 Process = .clr~new("System.Diagnostics.Process") --get access to public class

System.Diagnostics.Process

23

24 /* Method: Start

25 * Description: Starts the process based on already submitted information

26 * Argument: - ResetValues: calls class method "Reset" after starting the process

27 */

28 ::METHOD Start

29 EXPOSE ProcessStartInfo Application Filepath Process ResetValues

30 USE ARG ResetValues = .false

31

32 IF var("ProcessStartInfo") THEN

33 Process~clr.dispatch("Start", ProcessStartInfo) -- start process based on

ProcessStartInfo

34 ELSE IF var("Application") THEN

35 Process~clr.dispatch("Start", Application, Filepath) -- start process based on

Application and optionally Filepath

36 ELSE IF var("Filepath") Then

37 Process~clr.dispatch("Start", Filepath) -- start process based solely on Filepath

(uses default application)

38 ELSE

39 RETURN "ERROR: No values specified"

40

41 IF ResetValues <> .false THEN -- if the call to this method includes an argument, all

values will be reset after launching the process

42 self~Reset -- call to class method Reset

43

44 RETURN .true

45

46 /* Method: Reset

47 * Description: Resets all class attributes to their default values to launch a

different process in the same class instance

48 */

49 ::METHOD Reset

50 EXPOSE Application Filepath ProcessStartInfo

51 DROP application filepath processStartInfo -- drop the attributes

52

53 ::REQUIRES CLR.CLS

Listing 8: PROCESS.CLS

This coding example chains together a lot of commands separated by
message boxes, which have to be pressed in order to execute the next com-
mands. It also includes the ooRexx class "Process", which can be used for
easy process handling with one or more class instances. After instantiating
the public class "Process", one of the three attributes has to be de�ned or
else no process can be launched. Those three attributes are:

• Application: de�nes the Application to launch

• Filepath: de�nes the �le path to a �le to launch. If only this infor-
mation is available, Windows will start the process with the default
application based on the �le extension, e.g. Microsoft Word for .docx,

Acrobat Reader for .pdf and Notepad for .txt �les. These default
applications can be changed in the system settings.

• ProcessStartInfo: de�nes a set of values for the process. It has to be
an instance of the public class "System.Diagnostics.ProcessStartInfo"
and has to include at least the application or �le path as argument, like
in Line 54 of this coding example.

The class "Process" is set to public so it can be also used for other exam-
ples, it just needs to be included with the ::REQUIRES directive. Technically,
the inclusion of CLR.CLS in Line 64 is not needed since PROCESS.CLS already
requires CLR.CLS so therefore 06-process.demonstration.rxj has transitive ac-
cess to all the �les PROCESS.CLS has access to.

The last process to start in lines 54 to 61 includes a .NET class called
"ProcessStartInfo". It can have up to two arguments. The �rst one is the
�lepath and the second provides a set of commands to pass to the desired
application. Aside from the WindowStyle there are a lot of other Proper-
ties which can be set with ProcessStartInfo before starting the Process, for
example a (plain text or encrypted) password for applications which need
authentication before usage. It is also possible to start the process without
creating a new window - that can come in handy sometimes. [11]
When starting the application a text �le will be opened as seen in Figure 11.

Figure 11: Output of the �rst Notepad Application

Then, a message box gives the hint that a Website will be opened in the
preferred Webbrowser.

Figure 12: The �rst message box

The application opens https://google.at in the preferred browser in Figure
13.

Figure 13: https://google.at in favorite browser

Figure 14 shows another message box stating that the next web page will
be opened in Internet Explorer, regardless of the favorite browser setting.

Figure 14: The Second Message Box

Figure 15 opens the same web page in the Internet Explorer.

Figure 15: https://google.at in Internet Explorer

Finally, Figure 16 shows the notepad application in full screen mode.

Figure 16: Maximised Notepad Process

7 07-MAC.rxj

A Message Authentication Code is a tool for secure communication between
two subjects. It uses a hash function based on a secret key to generate a hash
value for a message. For the Message Authentication Code to work it is really
important that the secret key is only available to the sender and recipient
of the message. Once a key has been exchanged the communication with a
Message Authentication Code provides both integrity and authenticity.

1 /* File: 07-MAC.rxj

2 * Description: This application generates a hash string of a message based on a secret

key. Then, it stores both message and hash into a textfile and opens it afterwards as

a process with the default text editor program.

3 * Shows:

4 * - Encode a string into an array of Bytes

5 * - Generate a hash of a message based on a specific key

6 * - Convert an array of Bytes into a string

7 * - Usage of PROCESS.CLS to start a process

8 */

9

10 -- rgf: .file~separator statt "\"

11 -- rgf: aktuelles Verzeichnis kann man direkt in Rexx mit der directory()-Funktion

erhalten!

12

13 message = "This message needs a Message Authentication Code" -- an important message

14 key = "my123secret321key" -- this is a secret key, which only the sender and recipient

should have access to

15

16 encoding = .clr~new("System.Text.ASCIIEncoding")

17 keyByte = encoding~GetBytes(key) -- use ASCII encoding to code the key into several bytes

( a byte array )

18

19 hashkey = .clr~new("System.Security.Cryptography.HMACMD5", keyByte) -- get access to a MAC

md5 hash function class

20

21 messageBytes = encoding~GetBytes(message) -- returns the message as an array of Bytes

22

23 hashmessage = hashkey~ComputeHash(messageBytes) -- generates the hash in an array of Bytes

24

25 BitConverter = clr.import("System.BitConverter") -- get access to static class

System.BitConverter

26 MAC = BitConverter~toString(hashmessage) -- convert System.Byte[] to System.String

27

28 filedirectory = directory() -- this method returns the path of the current directory of

this application. This assures, that this application works on each and every system,

regardless of where the files are saved.

29 filename = "07-MAC.txt"

30 filepath = filedirectory || .file~separator || filename

31 file = .clr~new("System.IO.StreamWriter", filepath) -- get access to class

System.IO.StreamWriter

32 file~WriteLine(message) -- write message

33 file~~WriteLine~~WriteLine~~WriteLine~~WriteLine -- get 4 additional line breaks

34 file~Write("MAC for this message:" MAC) -- write MAC

35 file~Close

36

37 openResult = .Process~new -- get an instance of the Process Class

38 openResult~Filepath = filename -- set the filepath to our new filename. You could also use

variable "filepath" there - which is a reference to the absolute path + filename of

this file. But since it's in the same directory filename is sufficient.

39 openResult~Start -- start the process based on the given information (filepath)

40

41

42 ::REQUIRES PROCESS.CLS -- get access to Process class

Listing 9: A Message Authentication Code in ooRexx.NET

The .NET Framework has quite a few cryptography classes - all under the
namespace "System.Security.Cryptography". They include message encod-
ing, decoding, authentication and hash functions in general. Unfortunately,
some algorithms are already old and not recommended for using. It is best to
read the MSDN documentation on those classes carefully to avoid unsecure
applications.[12]
This example uses HMACMD5, a Hashed-Key Message Authentication Code
using MD5 Hashing. MD5 is a Message-Digest Algorithm with a 128 Bit/16
Byte hash. Therefore the length of the result of our Message Authentication
Code is a Byte array with exactly 16 Elements in it. This array is converted
to a string in Line 23 and then written into a �le with the message itself.
Finally, with the help of PROCESS.CLS the new text �le gets opened with
the default text editing program, in my case "notepad.exe".

Figure 17: Output of 07-MAC.rxj

8 08-WebClient.rxj

This coding example downloads the text document of an HTTP URL and
analyses it for each array element de�ned in Line 23.

1 /* File: 08-WebClient.rxj

2 * Description: This application downloads the frontend source code of an html page via

HTTP and analyses it based on the amount of html5 vs. html4.1 commands.

3 * Arguments: - URL including "http://" or "https://"

4 * Shows:

5 * - Setting up a WebClient to send or receive data through a URI

6 * - Defining a Rexx array with elements

7 * - Using Rexx arguments

8 * - Private Routine CountSubStrings to count substrings in a string

9 * - Using PROCESS.CLS to start a process

10 * - Comparing two integer values

11 */

12

13

14 WebClient = .clr~new("System.Net.WebClient") --get Access to the WebClient

15 IF arg(1)~length > 0 & ABBREV(arg(1), "http") -- if an argument was supplied while

starting this app AND this argument is a valid URL (starts with http at least)

16 THEN URL = arg(1) -- then set URL to that argument

17 ELSE URL = "https://wu.ac.at" -- else set it to the WU Homepage

18

19 SAY "analyzing web page at:" pp(url) "..."

20

21 htmlPage = WebClient~DownloadString(URL) -- download source code from given URL

22 htmlStructureElement = "<div"

23 html5StructureElements = .array~of("<article", "<header", "<nav", "<section", "<figure",

"<aside", "<footer")

24

25 htmlcount = CountSubStrings(htmlPage, htmlStructureElement) -- use private routine from

this file to count total amount of substrings

26 html5count = 0

27

28 DO oneElement OVER html5StructureElements

29 oneElementCount = CountSubStrings(htmlPage, oneElement) -- count all substrings for each

element of the html5StructureElements array

30 html5count += oneElementCount

31 SAY LEFT("amount of" oneElement, 18) ":" oneElementCount

32 END

33

34 s1 = "total amount of html5 Structure elements found:"

35 s2 = "pre html5 Structure elements found:"

36 maxLen = max(length(s1),length(s2)) -- get length of longer string

37 SAY s1~right(maxLen) html5count -- right adjust

38 SAY s2~right(maxLen) htmlcount -- right adjust

39 SAY

40

41 IF htmlcount > html5count THEN -- compare those values

42 SAY "The website" pp(url) "should get an html5 update!"

43 ELSE IF html5count>0 THEN -- html5 elements available?

44 SAY pp(url)": Great!"

45 ELSE

46 SAY pp(url)": ? Neither old HTML nor any HTML5 tags found, tsk!"

47

48

49

50 ::REQUIRES CLR.CLS

51

52 ::ROUTINE CountSubStrings PRIVATE

53 USE ARG haystack, needle

54 RegEx = .clr~new("System.Text.RegularExpressions.Regex", needle)

55 /* Although System Class "System.Text.RegularExpressions.Regex" has a constructor with

no Arguments, you cannot use it because it is PROTECTED.

56 Instead, 3 Constructors can be used: only needle, needle and RegexOptions or needle,

RegexOptions and Timespan. */

57

58 /* This example will count all non-overlapping occurences of the string defined in

"needle" in the string defined in "haystack" */

59 Matches = RegEx~Matches(haystack)

60 RETURN Matches~Count

Listing 10: Downloading and analysing HTML pages via HTTP in ooRexx.NET

The generated output of this application can be seen in Figure 18:

Figure 18: Output of 08-WebClient.rxj while analysing https://wu.ac.at

The private routine "CountSubStrings" starting in Line 52 creates an
regular expression class instance with the array element as argument. A
regular expression, also known as RegEx or RegExp, is a string with a speci�c
syntax which is used to search e�ciently through sometimes huge texts or to
specify how another string should look like. With regex pattern matching,
it makes it easy to create a rule that a string should start with an "A" and
end with "D" without restricting the letters between "A" and "D" - or that
it should or should not have a precise amount of characters. It is often used
for "search and replace" functions in many programming languages. Line
59 returns all occurrences of the variable needle in the variable haystack
and leaves the routine with the amount of found matches in Line 60. Since
html5 introduced quite a few new structural tags it is not su�cient to run
the private routine CountSubStrings only once - but for every tag and then

store the �nal results in the variables htmlCount and html5count. The �nal
output depends on whether there are more html5 structure elements than
divs on this web page or not. In case the reader wants to try a web page
with more html5 structure elements than html4, try http://html5demos.

com/contenteditable. In reality, no element can replace the good old "div"
tag counted in this example in the htmlCount variable. The html5 elements
are only an addition to the "div" tag since they allow to give elements of the
web page a better-named structure. [13]

9 09-clock.rxj

This coding example displays the current system time using .NET classes.

1 /* File: 09-clock.rxj

2 * Description: Continous output of the current time via .NET classes.

3 * Shows: - DateTime class: getting the current time.

4 * - Starting a CLRThread, which is a Thread and runs simultaneously to the

main code

5 * - Reading input keys from console

6 * - Exiting an application manually with the "Environment" class

7 */

8

9

10

11 SAY "Press any key to close this application."

12 SAY "----------------------------------------"

13

14 DateTime = .clr~new("System.DateTime")

15 .KeyPressThread~new~~start -- create an instance of KeyPressThread class, which is a

Thread and start it.

16

17 DO FOREVER -- starts an infinite loop

18 CALL SysSleep 1 -- wait 1 second for the time to change

19 currentTime = DateTime~Now -- DateTime~Now usually contains a timeString with that

format: "dd.mm.yyyy HH:mm:ss"

20

21 -- convert the timeString to only show the time, ooRexx~counterpart:

.dateTime~new~normalTime

22 SAY "The current time is" pp(currentTime~toString("HH:mm:ss"))

23 END

24

25

26 ::REQUIRES CLR.CLS --get ooRexx.NET support

27

28 /* Class: KeyPressThread

29 * Description: Checks continuously user input in the console. This class will then exit

the application using the "Environment" class.

30 */

31 ::CLASS KeyPressThread SUBCLASS CLRThread

32 ::ATTRIBUTE Console -- define attributes used in the class.

33 ::ATTRIBUTE Environment

34

35 ::METHOD init

36 EXPOSE Console Environment

37 Console = clr.import("System.Console") -- initialise console to use for the whole

class

38 Environment = clr.import("System.Environment") -- get access to static class

"Environment"

39

40 ::METHOD run -- by starting the CLRThread, this method will be run automatically

41 EXPOSE Console Environment -- get Access to Console and Environment

42

43 DO FOREVER -- starts an infinite loop

44 IF Console~ReadKey <> .nil THEN -- Method "ReadKey" saves user input. If nothing was

pressed, it will return the NIL Object

45 DO

46 Environment~Exit(0) -- leave application

47 EXIT -- leave method

48 END

49 END

Listing 11: Displaying the current time as it changes in ooRexx.NET

Figure 19: Output of 09-clock.rxj

This application runs an in�nite loop where it retrieves the current sys-
tem time, which consists of the date and the time. Since we only need the
time, the "toString" method allows for an argument to display only the �elds
that are needed - in this example the hour, minute and second. The syntax
for the hour is either HH or hh. The di�erence between those is that small
caps hh use an 12-hour clock, while large caps HH use an 24-hour clock. This
method of retrieving the time may not be the most e�cient one, since a lot
of calculations have to be made in order to display the current time. On the
other hand it is very secure and therefore advisable for server applications.
Try the following: change the time in your system settings while running this
application. As a result the displayed time will also immediately change! Un-
fortunately, due to millisecond rounding and lag it is possible that certain
seconds will get left out like [20:22:47] in Figure 19. This is quite unfortunate,
but it can happen. As a countermeasure the author suggests to implement
one of two things:

• Retrieve the system time only every 5 seconds or so. For the other sec-
onds simply count an integer up. This should not create any signi�cant
lag whatsoever.

• Reduce SysSleep time of Line 18 by a few milliseconds.

The class "KeyPressThread" is a subclass of the class "CLRThread", which
allows it to run asynchronously. It waits for the user to press a key and then
exits the application. But �rst it de�nes two attributes in Lines 32 and 33,
which is actually not necessary for the functionality of this application - but
still it is easier to read if the class gets expanded.

10 10-gui.introduction.rxj

This application introduces Windows Forms, which can be used to create
Graphical User Interfaces (GUIs) using the .NET Framwork.

1 /* File: 10-gui.introduction.rxj

2 * Description: This application opens a Windows Form including a label with text.

3 * Shows:

4 * - A Graphical User Interface (GUI) in ooRexx.NET

5 * - Setting the title of a Form

6 * - Disallowing maximization and minimization of the Form

7 * - Attaching an Icon to a Form

8 * - Setting a Startposition for the Form to the center of the screen

9 * - Creating a simple label with a text

10 * - Setting the Location of the Label

11 * - Changing font family, font size and font style of a Label

12 * - Adding the label to the controls of the Form

13 * - Starting the Form, therefore displaying it.

14 */

15

16

17 form = .clr~new("System.Windows.Forms.Form") -- get an instance of the Form class of

namespace "System.Windows.Forms"

18 form~Text = "Hello World" -- set the Caption of the Titlebar

19 form~maximizeBox = .false -- do not allow to maximize the form - therefore, do not even

display the buttons for maximization/minimization

20 form~minimizeBox = .false -- do not allow to minimize the form

21 form~Width = 400 -- set the width of the form to 400px

22 form~Height = 300 -- set the height of the form to 300px

23

24 SystemIcons = clr.import("System.Drawing.SystemIcons")

25 icon = .clr~new("System.Drawing.Icon", SystemIcons~Shield, 48, 48) -- load the enumation

"Shield" from the set of system icons in 48x48px size as an icon

26 form~Icon = icon -- attach that icon to the form

27

28 FormStartPosition = clr.import("System.Windows.Forms.FormStartPosition") -- import static

enumeration "FormStartPosition"

29 form~startPosition = FormStartPosition~CenterScreen -- set the starting position of the

form to the center of the screen

30

31 label = .clr~new("System.Windows.Forms.Label") -- get an instance of the Label class

32 label~Text = "Hello, my beloved World - from ooRexx.NET" -- set the text to "Hello World"

33 label~Location = .clr~new("System.Drawing.Point", 25, 40) -- set the location to 40px from

top and 25px from left corner of the form

34 label~AutoSize = .true -- resize this label from default label size to real label size

35

36 FontStyle = clr.import("System.Drawing.FontStyle") -- get access to enumation FontStyle

37 font = .clr~new("System.Drawing.Font", "Verdana", 10, FontStyle~Italic)

38 label~Font = font -- set the Font of the label to font family "Verdana" with font size

10px and font style "Italic"

39

40

41 form~Controls~Add(label) -- add the label to the Controls of the form

42 SAY "close the dialog to end program"

43 form~showDialog -- show the form

44

45 ::REQUIRES CLR.CLS -- get ooRexx.NET Support

Listing 12: Displaying a Windows Form using ooRexx.NET

In ooRexx.NET, one just has to create an instance of the Form Class and
add Controls to it. There are many di�erent controls in .NET, for example
labels, buttons or a progress bar. Basically anything, which is used for user
interaction in a GUI can be added to the form as a control object. Finally,
to display the dialog, there are 2 ways to achieve that:

• Use one of the methods "ShowDialog" or "Show" from the Form Class.
There is no di�erence in functionality between them.

• Use the method "Run" from the Application class with
"System.Windows.Forms" namespace

The opened form can be seen in Figure 20.

Figure 20: Output of 10-gui.introduction.rxj

11 11-drawing.rxj

This application demonstrates image and bitmap handling in ooRexx.NET.
It �rst opens the image as a bitmap from the "images" directory and then
retrieves the �le size and other properties of the image. After displaying it
in a "PictureBox" the application crops the image by 50 % .

1 /* File: 11-drawing.rxj

2 * Description: This application loads an image from the "images" subfolder, outputs a few

interesting properties and then crops the image by 50% and saves it under a different

file name.

3 * Shows:

4 * - Loading a Bitmap from the image path

5 * - Accessing various Bitmap properties

6 * - Retrieving image file size with FileInfo Class

7 * - Adding thousand points to an integer for better overview

8 * - Applying a new image size to an image

9 * - Displaying an image in a form

10 */

11

12 imagePath = "images/html5.jpg"

13 bitmap = .clr~new("System.Drawing.Bitmap", imagePath) -- load the image as a Bitmap

14 width = bitmap~width

15 height = bitmap~height

16 dpi = bitmap~VerticalResolution

17 pxFormat = bitmap~PixelFormat

18

19 SAY "Get properties of the loaded Image"

20

21 SAY "Width:" pp(width"px") -- pp = pretty print, inserts "[" before and "]" after the

argument

22 SAY "Height:" pp(height"px")

23 SAY "Dots per Inch (DPI):" pp(dpi"dpi")

24 SAY "Pixel format:" pp(pxformat)

25 /* now let's retrieve the file size of this image */

26 filesize = .clr~new("System.IO.FileInfo", imagepath)~length -- returns filesize in Bytes

27 --SAY "File size:" pp(filesize~toString("N0") "Bytes") -- use number 0 default formatting

to add thousand points while converting Integer64 to String

28 SAY "File size:" pp(filesize "Bytes")

29 SAY "----------------------------------------"

30

31 SAY "now, let's resize and display that image"

32 size = .clr~new("System.Drawing.Size", 960, 540) -- half size

33 bitmap = .clr~new("System.Drawing.Bitmap", bitmap, size) -- load a new bitmap from the old

bitmap but use the new size

34 form = .clr~new("System.Windows.Forms.Form") -- a Form represents a window as a user

interface

35 form~size = size -- set the width and height of the form to the width and height of the

(new) bitmap

36

37 pictureBox = .clr~new("System.Windows.Forms.PictureBox") -- create a new PictureBox, which

is used to display images in a form

38 pictureBox~image = bitmap -- insert our bitmap into the PictureBox

39 pictureBox~size = size -- also set the the width and height of the PictureBox to the size

of our bitmap

40

41 SAY "new Width:" pp(bitmap~width) -- output width and height of the bitmap to see, if

resizing was successfull

42 SAY "new Height:" pp(bitmap~height)

Page 36



Coding Examples 11-drawing.rxj

43

44 form~controls~add(pictureBox) -- add the PictureBox to the form

45 SAY "close the dialog to proceed..."

46 form~showDialog -- show the form

47

48 SAY "save new image? (Y/N)"

49 PULL save -- retrieve user input in console and save it as variable "save"

50 IF save = "Y" THEN DO -- since we used command "PULL" instead of "PARSE PULL", small caps

user input will be transformed to upper case. Therefore this command will return .true

even if the user typed in small caps "y".

51 newImagePath = "11-newImage.jpg"

52 bitmap~save(newImagePath) -- save the bitmap as filename supplied as argument

53 SAY "0A"x"Image successfully saved" -- insert line feed before actual Output

54

55 newfilesize = .clr~new("System.IO.FileInfo", newImagePath)~length -- returns new

filesize in Bytes

56 -- turn the number into a CLR/.Net System.Decimal value and use the formatting

capabilities of its ToString() method

57 d=clr.box("decimal",newfilesize) -- box as a System.Decimal (range should be large

enough for all primitive value types)

58 SAY "new File size:" pp(newfilesize "Bytes") "or formatted:" pp(d~toString("N0"))

"Bytes"

59 END

60 ELSE EXIT 0

61

62

63 ::REQUIRES CLR.CLS -- get ooRexx.NET Support

Listing 13: Image handling with ooRexx.NET

Figure 21: Console Output of 11-drawing.rxj

Figure 22: Form Output of 11-drawing.rxj

After the image gets resized, the application outputs the new �le size
converted to a string with thousand points so it is easier to read for human
users. The .NET Framework supports quite a few standard numeric format
strings. Here is a list of the most useful ones:

• "C", en-US stands for "Currency" and uses the country code of the
second argument to represent a value as a currency. The use of this
format speci�er for the number "132.456" would result in "$123.46"".
For the German country code, use de-DE.

• "D10" formats the provided integer to the speci�ed length. So this
number speci�er would result in "0000123456" for number "123456".
This is useful, if a database requires a special number format.

• "N2" formats to the number of decimal places. This can be extended
by the country code in the second argument.

• "P" for displaying percentages. 0 = 0 %, 1 = 100 %

• "H" formats a decimal number as an hexadecimal value. Open Object
Rexx uses D2X(wholenumber) Function for that.

[14]

12 12-save�le.rxj

This application shows an easy implementation of a "save �le" dialog using
forms.

1 /* File: 12-savefile.rxj

2 * Description: This application opens a form with a textarea and a save button to save

the text as a text file.

3 * Shows:

4 * - Customizing a textbox

5 * - Using Threads

6 * - A dialog to save files

7 * - Filtering file names and file types based on regular expressions

8 */

9

10

11 form = .clr~new("System.Windows.Forms.Form") -- get an instance of the Form class

12 form~AutoSizeMode = GrowAndShrink -- set the attribute "AutoSizeMode" to enumeration

"GrowAndShrink". This allows the form to automatically adjust to the elements.

13 form~text = "Save Text"

14 form~width = 610

15 form~height = 545

16

17 textbox = .clr~new("System.Windows.Forms.TextBox") -- get an instance of the TextBox class

18 textbox~Multiline = .true -- allows for multiple lines

19 textbox~AcceptsReturn = .true -- this attribute allows line breaks

20 textbox~AcceptsTab = .true -- this attribute allows tabs

21

22 textbox~size = .clr~new("System.Drawing.Size", 500, 500) -- set the size of the textbox to

500x500px with

23

24 font = .clr~new("System.Drawing.Font", "Consolas", 14) -- get an instance of the Font

class and set it to font family "Consolas" with font size 14pt

25 textbox~font = font -- apply our font instance to the textbox, therefore setting the text

of the textbox to Consolas/14

26

27 startButton = .clr~new("System.Windows.Forms.Button")

28 startButton~Text = "Save" -- set property "Text" to string "Save"

29

30 contentPane = .clr~new("System.Windows.Forms.FlowLayoutPanel") -- represent a

FlowLayoutPanel to lay out the content horizontally

31 contentPane~AutoSize = .true -- set property "AutoSize" to boolean true

32 contentPane~AutoSizeMode = GrowAndShrink -- set property "AutoSizeMode" to enumeration

value "GrowAndShrink"

33 form~Controls~Add(contentPane) -- add the contentPane to the form

34

35 contentPane~Controls~Add(textbox) -- add textbox to the contentPane (therefore adding it

to the form too)

36 contentPane~Controls~Add(startButton) -- add the button to the contentPane (therefore

adding it to the form too)

37

38 userData = .directory~new -- create a new Rexx Directory

39 userData~TextBox = textbox -- add the textbox to the directory

40 userData~startButton = startButton -- add the button to the directory

41

42 mouseEventHandler = clr.createEventHandler(.MouseEventHandler~new(userData)) -- create

new event handler from ooRexx class "MouseEventHandler" with the directory "userData"

as parameter

43 startButton~Click += mouseEventHandler -- register event handler to "Click" event

44

45 application = clr.import("System.Windows.Forms.Application") -- get reference to static

class "System.Windows.Forms.Application"

46 application~Run(form) -- invoke method "Run" to start the form

47

48

49 ::REQUIRES CLR.CLS -- get ooRexx.NET support

50

51 /* Class: MouseEventHandler

52 * Description: Instantiates the ooRexx "Save" class with the passed userData as

parameter, which is a thread, therefore starting it with the "Start" method.

53 */

54 ::CLASS MouseEventHandler

55

56 /* Method: init

57 * Description: This constructor saves the passed arguments in the class.

58 * Arguments: - userData: an ooRexx directive, which can reference several form

elements

59 */

60 ::METHOD init

61 EXPOSE userData

62 USE ARG userData

63

64 /* Method: invoke

65 * Description: Called by .NET when the registered event is triggered.

66 * Arguments: - caller: the object the event handler was registered to

67 * - mouseEventArgs: the event arguments passed by the caller

68 */

69

70 ::METHOD invoke

71 EXPOSE userData

72 USE ARG caller, mouseEventArgs

73 .Save~new(userData)~start -- start the "Save" class, which is a Thread

74

75

76

77 ::CLASS Save SUBCLASS CLRThread

78 ::METHOD init

79 EXPOSE userData

80 USE ARG userData

81

82 /* Method: run

83 * Description: This method will get executed upon start of the thread by invoking

"start" method of the CLRThread class, which is a superclass of this class.

84 */

85 ::METHOD run

86 EXPOSE userData -- expose must be the first command after invocation of a

method

87

88 textbox = userData~TextBox -- get a (shorter) reference to the textbox instance

89 SaveFileDialog = .clr~new("System.Windows.Forms.SaveFileDialog") -- get an instance of

the SaveFileDialog class

90 SaveFileDialog~Filter = "all files|*.*" -- Syntax: [Label]|[Regular Expression]

91 SaveFileDialog~FileName = "12-savefile.txt" -- this provides a default value as the

file name, which can be overwritten in runtime

92

93 DialogResult = clr.import("System.Windows.Forms.DialogResult") -- import static

DialogResult class in order to check where the user clicked later on

94 SaveFileDialog.Result = SaveFileDialog~ShowDialog -- show the dialog and save where

the user clicked in the variable "SaveFileDialog.Result"

95

96 IF SaveFileDialog.Result~equals(DialogResult~OK) THEN DO

97 FileStream = .clr~new("System.IO.StreamWriter", SaveFileDialog~OpenFile) -- open a

filestream based on the filename/filepath+filename from the Save Dialog

98 FileStream~Write(textbox~Text) -- write the text from the textbox into that

filestream file

99 FileStream~Close

100 MessageBox = clr.import("System.Windows.Forms.MessageBox")

101 MessageBox~Show("finished saving process", "success") -- show the user that saving

the file was successfull

102 END

Listing 14: A save �le dialog ooRexx.NET

Figure 23: The form of 12-save�le.rxj

The "SaveFileDialog" opens when clicking on the "Save" button.

Figure 24: The save �le dialog of 12-save�le.rxj

If saving was successful, the user gets noti�ed via a message box in Figure
25.

Figure 25: The con�rmation dialog which pops up after saving the �le in 12-save�le.rxj

Figure 26: An alternative output when disabling attribute "multiline" of the form in
12-save�le.rxj

The dialog in Figure 24 basically asks the user to select a folder and
provide a �le name. After clicking on the "save" button of the "save �le"
dialog it returns the absolute path of the �le name without creating that
�le in Line 94. What happens with that �le path is up to the application.
12-save�le.rxj uses the given �le path and �le name to create a new text �le
using the SteamWriter class of System.IO namespace in Line 98.
Interestingly, the "Multiline" attribute de�ned in Line 18 is not bound to
the height of the TextBox. Disabling it will result in a single-lined text box
absolutely ignoring the height attribute. However, even when setting the
size of the TextBox with the "size" attribute, not "width" and "height", a
disabled "multiline" attribute will have no impact on the width.
The Rexx constructor used in both ooRexx classes "MouseEventHandler" and
"Save" is a method called "init", which stands for "initialize" and is called
directly after instantiating the class. Usually there exist several constructors
with di�erent parameters, so that the reader has a choice whether the reader
wants to provide information about that class at time of class creation or
not.

13 13-load�le.rxj

This application shows the "OpenFileDialog" where the user can select a text
�le he wants to open.

1 /* File: 13-loadfile.rxj

2 * Description: This application opens a form with a textarea and a "load" button to load

a text file and display it in the textarea.

3 * Shows:

4 * - Using a RichTextBox in contrast to the standard TextBox

5 * - Enabling shortcuts in a (rich)textbox

6 * - The Form.Load Event in ooRexx.NET used for executing commands after

loading the form

7 * - Filtering file names and file types based on regular expressions

8 */

9

10

11 form = .clr~new("System.Windows.Forms.Form")

12 form~AutoSizeMode = GrowAndShrink

13 defaultFormText = "Load file demonstration"

14 form~text = defaultFormText

15 form~width = 610

16 form~height = 545

17 textbox = .clr~new("System.Windows.Forms.RichTextBox")

18 textbox~Multiline = .true

19 textbox~AcceptsTab = .true

20 textbox~ShortcutsEnabled = .true -- this attribute allows amongst other things ctrl+a to

select everything

21 textbox~size = .clr~new("System.Drawing.Size", 500, 500)

22 font = .clr~new("System.Drawing.Font", "Consolas", 14)

23 textbox~font = font

24

25 startButton = .clr~new("System.Windows.Forms.Button")

26 startButton~Text = "Load again" -- set property "Text" to string "Load again"

27

28 contentPane = .clr~new("System.Windows.Forms.FlowLayoutPanel")

29 contentPane~AutoSize = .true -- set property "AutoSize" to boolean true

30 contentPane~AutoSizeMode = GrowAndShrink -- set property "AutoSizeMode" to enumeration

value "GrowAndShrink"

31 form~Controls~Add(contentPane)

32

33 contentPane~Controls~Add(textbox)

34 contentPane~Controls~Add(startButton)

35

36 userData = .directory~new

37 userData~form = form -- add the form to the directory

38 userData~TextBox = textbox

39 userData~startButton = startButton

40 userData~defaultFormText = defaultFormText -- also include the default form text of this

application to prevent redundancy

41 mouseEventHandler = clr.createEventHandler(.MouseEventHandler~new(userData)) -- create

new event handler from ooRexx class "MouseEventHandler"

42 startButton~Click += mouseEventHandler -- register event handler to "Click" event

43 form~Load += mouseEventHandler -- Occurs before a form is displayed for the first time.

This will open the "load file" dialog when starting the application.

44

45 application = clr.import("System.Windows.Forms.Application") -- get reference to static

class "System.Windows.Forms.Application"

46 application~Run(form) -- invoke method "Run", which starts an application message loop

47

48

49 ::REQUIRES CLR.CLS -- get ooRexx.NET support

50

51

52 ::CLASS MouseEventHandler

53

54 /* Method: init

55 * Description: Constructor which saves the passed arguments in the class.

56 * Arguments: - progressBar: the progress bar which is to be modified

57 * - startButton: the start button which is to be modified

58 */

59 ::METHOD init

60 EXPOSE userData

61 USE ARG userData

62

63 /* Method: invoke

64 * Description: Called by .NET when the registered event is triggered.

65 * Arguments: - caller: the object the event handler was registered to

66 * - mouseEventArgs: the event arguments passed by the caller

67 */

68 ::METHOD invoke

69 EXPOSE userData

70 USE ARG caller, mouseEventArgs

71 .Save~new(userData)~start

72

73

74

75 ::CLASS Save SUBCLASS CLRThread

76 ::METHOD init

77 EXPOSE userData

78 USE ARG userData

79

80 /* Method: run

81 * Description: Executed upon start of the thread (by invoking "start").

82 */

83 ::METHOD run

84 EXPOSE userData

85 fileContent = loadFile()

86 IF fileContent <> .false THEN DO

87 userData~textbox~Text = fileContent[1] -- insert the file content to the textarea

88 userData~form~Text = userData~defaultFormText || ":" fileContent[2] -- add the file

name (including file extension) to the title of the form

89 END

90

91 /*

92 * ROUTINE: loadFile

93 * Description: opens a "open file" dialog to select a text or rexx file

94 * Returns: either an array with file content and file name or the ooRexx value

".false" / 0

95 */

96

97 ::ROUTINE loadFile

98 OpenFileDialog = .clr~new("System.Windows.Forms.OpenFileDialog")

99 DialogResult = clr.import("System.Windows.Forms.DialogResult")

100 OpenFileDialog~Filter = "text files|*.txt|rexx files|*.rex" -- Syntax: Label|Regular

Expression

101 OpenFileDialog~Title = "Open a file" -- set the title of the OpenFileDialog

102

103 OpenFileDialog.Result = OpenFileDialog~ShowDialog

104 IF OpenFileDialog.Result~equals(DialogResult~OK) THEN DO

105 FileStream = .clr~new("System.IO.StreamReader", OpenFileDialog~OpenFile)

106 FileContent = FileStream~ReadToEnd

107

108 IF FileContent <> .nil THEN DO

109 FileStream~Close -- only close the filestream if the content is OK

110 FileName = OpenFileDialog~SafeFileName -- retrieve only the file name without file

path

111 returnArray = .array~of(FileContent, FileName)

112 RETURN returnArray

113 END

114 ELSE

115 SAY "FileContent is empty :-( Try again!"

116 END

117 ELSE

118 RETURN .false

Listing 15: A open �le dialog ooRexx.NET

Figure 27: The form of 13-load�le.rxj also opens a "open �le" dialog

Figure 28: A huge bene�t of RichTextBoxes: They also display http links as real
clickable links

This application o�ers to open a �le directly at the start using a "Form.Load"
event from the form class.
In the previous example there was a huge disadvantage of the "TextBox"
class. It would not allow for shortcuts like ctrl+a to select the whole text.
That is why we use the "RichTextBox" class in this example, because it
has always multiple lines - so setting the "multiline" attribute will not
have any impact on the output of the application. Therefore there is no
"AcceptsReturn" attribute in the RichTextBox class, which allows the user
for creating line breaks. Setting it to a boolean (or any other) value like in
the previous example will always result in a runtime error when executing
the application. A complete list of all available shortcuts can be found on
the corresponding Microsoft website - see [15, 16]

14 14-menu.rxj

This application opens a form with a menu bar including multiple menu
items.

1 /* File: 14-menu.rxj

2 * Description: This application opens a form with a menu bar, a menu inside a menu and a

button at the bottom of the form.

3 * Shows:

4 * - Using a FlowLayoutPanel to group together Control Elements.

5 * - The "Dock" attribute to change the position of Control Elements

dynamically.

6 * - Using the "caller" attribute when invoking event handlers to determine

the clicked element

7 * - Creating a border, background color and background image for form

elements

8 * - Creating a Menu

9 * - Creating a SubMenu

10 * - How to assign a shortcut to a menu item.

11 */

12

13 form = .clr~new("System.Windows.Forms.Form")

14 form~text = "Menu Class"

15 form~AutoSizeMode = GrowOnly

16 FormStartPosition = clr.import("System.Windows.Forms.FormStartPosition")

17 form~StartPosition = FormStartPosition~CenterScreen -- start the form in the center of the

screen

18

19 endButton = .clr~new("System.Windows.Forms.Button")

20 endButton~Text = "Exit Application" -- set property "Text" to string "Exit Application"

21 endButton~Width = 275

22

23 contentPane = .clr~new("System.Windows.Forms.FlowLayoutPanel") -- represent a

FlowLayoutPanel to lay out the content horizontally

24 contentPane~width = form~width -- set the width of the contentPane to the width of the

form

25 contentPane~height = 30 -- default height of a button is 24px, so set the contentPane to a

little more than that: 30px

26

27 BorderStyle = clr.import("System.Windows.Forms.BorderStyle")

28 contentPane~BorderStyle = BorderStyle~FixedSingle -- add a single dimensional border to

the contentPane

29

30 DockStyle = clr.import("System.Windows.Forms.DockStyle")

31 contentPane~Dock = DockStyle~Bottom -- move the contentPane including all its components

(the button) to the bottom of the form by docking it

32

33

34 Color = clr.import("System.Drawing.Color") -- get a reference of the static

System.Drawing.Color class

35 contentPane~BackColor = Color~LightGray -- set the background color of the contentpane to

a predefined light gray color

36 form~BackColor = Color~White

37 endButton~BackColor = Color~Lavender -- Lavender seems to be an even lighter gray

38

39

40 bgImage = .clr~new("System.Drawing.Bitmap", "images/oorexx_logo.gif") -- import the

oorexx_logo image as a bitmap

41 form~backgroundImage = bgImage -- apply the imported background image to the form

42 bgImageLayout = clr.import("System.Windows.Forms.ImageLayout") -- get a reference to the

static System.Windows.Forms.ImageLayout class

43 form~backgroundImageLayout = bgImageLayout~Center -- set the background image to the

center of the form

44

45 contentPane~Controls~Add(endButton) -- add the button to the contentPane

46 form~Controls~Add(contentPane) -- add the contentPane to the form

47

48 application = clr.import("System.Windows.Forms.Application") -- get reference to static

class "System.Windows.Forms.Application"

49

50 userData = .directory~new

51 userData~form = form

52 userData~application = application

53

54 mouseEventHandler = clr.createEventHandler(.MouseEventHandler~new(userData)) -- create

new event handler from ooRexx class "mouseEventHandler"

55 endButton~Click += mouseEventHandler -- register event handler to "Click" event

56

57 menu = .clr~new("System.Windows.Forms.MainMenu") -- get an instance of the MainMenu class

58 menuItemFile = .clr~new("System.Windows.Forms.MenuItem", "File") -- create a new MenuItem

with the title "File"

59 menuItemHelp = .clr~new("System.Windows.Forms.MenuItem", "Help") -- create a new MenuItem

with the title "Help"

60 shortcut = clr.import("System.Windows.Forms.Shortcut") -- get access to static class

"System.Windows.Forms.Shortcut"

61 menuItemHelp~Shortcut = shortcut~F1 -- give menu item "Help" the shortcut "f1"

62 menuItemHelp~Click += clr.createEventHandler(.showHelp~new) -- if this menu item gets

activated, create an event handler and get an instance of the showHelp class

63

64 menuItemClose = .clr~new("System.Windows.Forms.MenuItem", "Close")

65 menuItemClose~Shortcut = shortcut~AltF4

66 menuItemClose~Click += clr.createEventHandler(.MouseEventHandler~new(userData)) --

register event handler to "Click" event. This menu item does the same as the endButton

67

68 openFile = .clr~new("System.Windows.Forms.MenuItem", "Open File") -- this is a menu item

to demonstrate menus for menus, e.g. submenus

69 saveFile = .clr~new("System.Windows.Forms.MenuItem", "Save File")

70 menuItemFile~MenuItems~~Add(openFile)~~Add(saveFile)

71

72 possibilities = .directory~new

73 possibilities~open = openFile

74 possibilities~save = saveFile

75

76 openFile~Click += clr.createEventHandler(.WhoAmI?~new(possibilities))

77 saveFile~Click += clr.createEventHandler(.WhoAmI?~new(possibilities))

78

79 menu~MenuItems~Add(menuItemFile)

80 menu~MenuItems~Add(menuItemHelp)

81 menu~MenuItems~Add(menuItemClose)

82 form~Menu = menu -- set our menu as the Menu of the form

83

84

85 application~Run(form) -- invoke method "Run", which starts an application message loop

86

87

88

89 ::REQUIRES CLR.CLS

90

91 /* Class: MouseEventHandler

92 * Description: Creates an instance of the ooRexxx "ExitApplication" class and starts it

93 * Argument: A directory with the form, which has to be closed.

94 */

95

96 ::CLASS MouseEventHandler

97 ::METHOD init

98 EXPOSE userData

99 USE ARG userData

100

101 ::METHOD invoke

102 EXPOSE userData

103 USE ARG caller, mouseEventArgs

104 .ExitApplication~new(userData)~~start

105

106

107 /* Class: ExitApplication

108 * Description: This class closes a form.

109 * Argument: A directory with the form, which has to be closed.

110 */

111

112 ::CLASS ExitApplication SUBCLASS CLRThread

113 ::METHOD init

114 EXPOSE userData

115 USE ARG userData

116

117 ::METHOD run

118 EXPOSE userData

119

120 MessageBox = clr.import("System.Windows.Forms.MessageBox") -- get access to static

class "MessageBox" of namespace "System.Windows.Forms"

121 MessageBoxButtons = clr.import("System.Windows.Forms.MessageBoxButtons") -- get access

to a set of enumerations

122

123 MessageBox.Result = MessageBox~show("Do you really want to close this applicaton?",

"Exit application", MessageBoxButtons~YesNo) -- display MessageBox and save result in

Rexx variable "MessageBox.Result"

124 DialogResult = clr.import("System.Windows.Forms.DialogResult") -- get a reference to

DialogResult enumerations

125

126 IF MessageBox.Result~equals(DialogResult~Yes) THEN DO

127 form = userData~form

128 form~Close

129 END

130

131 /* Class: showHelp

132 * Description: Creates an instance of the ooRexxx "runHelp" class, which is a Thread,

and starts it

133 */

134

135 ::CLASS showHelp

136 ::METHOD invoke

137 USE ARG caller, mouseEventArgs

138 .runHelp~new~start

139

140

141 /* Class: runHelp

142 * Description: This class opens a MessageBox with a hint

143 */

144

145 ::CLASS runHelp SUBCLASS CLRThread

146 ::METHOD run

147 MessageBox = clr.import("System.Windows.Forms.MessageBox")

148 MessageBox~show('hint: press "f1" to access this help box or "alt + f4" to close the

application', "Help")

149

150

151 /* Class: WhoAmI?

152 * Description: This class validates the clicked element based on the parameter and

outputs a MessageBox with the name of the clicked element.

153 * Argument: A directory with all available menu items whose click event handler

target is the "WhoAmI?" class.

154 */

155

156 ::CLASS WhoAmI?

157 ::METHOD init

158 EXPOSE bothItems MessageBoxTitle

159 USE ARG bothItems

160 MessageBoxTitle = "WhoAmI: Result" -- set the title to be used later on

161

162 ::METHOD invoke

163 EXPOSE bothItems MessageBoxTitle

164 USE ARG caller, mouseEventArgs

165 saveItem = bothItems~save -- retrieve the reference to the menu item "save"

166 openItem = bothItems~open

167

168 MessageBox = clr.import("System.Windows.Forms.MessageBox")

169 IF caller~toString = saveItem~toString THEN -- convert both objects into a string

representation and check for equality

170 MessageBox~show("You clicked on the ""save"" menu item!", MessageBoxTitle)

171 ELSE IF caller~toString = openItem~toString THEN

172 MessageBox~show("You clicked on the ""open"" menu item!", MessageBoxTitle)

Listing 16: Using menus and submenus in ooRexx.NET

After executing this application this form will open providing a menu bar,
a background image and a button to exit the application as seen in Figure
29.

Figure 29: This image shows the form of the example 14-menu.rxj

The message box from Figure 30 will pop up if "File > Save" gets clicked.
The clicked element gets identi�ed by the "caller" argument in Lines 169-172.

Figure 30: This message box invoked in the WhoAmI? class.

The message box from Figure 31 will pop up if "File > Open" gets clicked.

Figure 31: Another message box invoked in the WhoAmI? class.

The message box in �gure 32 provides some hints for this application. It
gets activated when the user clicks on "Help" in the menu bar or presses "f1"
on the keyboard.

Figure 32: The output of the class "runHelp" in Lines 145-148

Finally, this message box con�rms the "Exit Application" button at the
bottom and the "Exit" menu item in the menu. The application will close
once the user con�rms by pressing on the "Yes" button

Figure 33: The output of the class "ExitApplication" in Lines 112-129.

This application primarily demonstrates menus and shows how much code
they actually need to perform well. Including a menu bar in an application
are just two lines of code - the �rst being the import of the main menu bar
itself, see Line 57 and the second one is embedding the menu bar in the form.
Then, each menu item requires at least two additional lines of code as shown
in Line 58 and 79. Of course, menu items have a speci�c purpose, for ex-
ample calling a routine or instantiating a class. That requires adding event
handlers which add several lines of code. Aside from the many attributes
that can be changed in the menu bar and the menu items it can get really
confusing when creating applications with one or even more menus. The au-
thor therefore advises to use additional �les to keep the structure inside the
source code of an application. It is possible to include other Rexx �les in an
application with the ::REQUIRES directive. This way, menu bars can stay the
way they are and yet the programmer has no problem to keep an overview
over his code.
Another useful and easy-to-implement thing this application shows are lines
40 to 43, which add a background image to the form. First, the image gets
imported as a bitmap from the �le path. The "BackgroundImageLayout" prop-
erty of the form class determines where the image will be displayed. Unfor-
tunately, this property does not allow for much creativity, since it expects a
static enumeration value of type "System.Windows.Forms.ImageLayout". The
possibilities of this enumeration are quite limited as it for example does not
allow for the background image to display in the upper right corner of the

form. Actually, the background image cannot be displayed in any corner
except the upper left, which is equivalent to a Point with the coordinates 0
and 0. So this is quite a limitation but there are multiple solutions for this
problem. A background image can be attached to most of the form elements
in .NET. Therefore it is possible to set a content pane with the background
image to dock to the upper right corner of the form. Alternatively, one could
load the image into a new bitmap with the exact size of the form, just like
in example 11-drawing.rxj on page 37.
However, images are a great addition to any application and are worth their
troubles, just like menus are. [17]

15 15-text.to.speech.rxj

This application shows one of the many great advantages of the .NET frame-
work, a built-in speech synthesis engine. This feature generates audio output
based on textual input by analysing linguistic components like phasing, in-
tonation and duration.

1 /* File: 15-text.to.speech.rxj

2 * Description: This application first reads a default text, then reads the content of

"02-textfile.txt" word by word ignoring the first line.

3 * Shows:

4 * - Adding an Assembly

5 * - Regulating volume and rate of the audio output

6 * - Reading a string

7 * - Extracting words of a textfile

8 */

9

10 CALL clr.addAssembly "System.Speech" -- load System.Speech Assembly to use all

System.Speech.* classes

11

12 Console = clr.import("System.Console")

13 SpeechSynthesizer = .clr~new("System.Speech.Synthesis.SpeechSynthesizer") -- get an

instance of the SpeechSynthesizer class to use all text-to-speech (TTS) functions

14 SpeechSynthesizer~SetOutputToDefaultAudioDevice

15 SpeechSynthesizer~Volume = 100 -- regulate the volume of the voice

16 SpeechSynthesizer~Rate = -1 -- regulate the speed between -10 and 10.

17

18 TextToRead = "This is the default text. If you want me to read something else, just open

this application with the text as argument. You will not regret it."

19 IF arg(1)~length > 0 THEN TextToRead = arg(1)

20

21 Console~WriteLine(TextToRead)

22 SpeechSynthesizer~Speak(TextToRead)

23

24 CALL SysSleep 1 -- wait a second

25

26 filepath = "02-textfile.txt"

27 mystream = .stream~new(filepath) -- load a new filestream using ooRexx STREAM Class

28 mystream~~linein -- ignore first line, which is the heading with a lot of tildes

29 SpeechSynthesizer~Rate = 1 -- speed the rate of the speaker up a little bit

30

31 word = "" -- set the variable "word" to a 0 length string

32

33 SAY "------------------------------------"

34 Console~WriteLine("Reading" filepath || ":")

35 SpeechSynthesizer~Speak("Reading")

36 SpeechSynthesizer~Rate = -6 /* change the rate to -6 because the filepath consists of a

lot of points and slashes.

37 Still, sounds weird because she says "zero two to textfile" instead of "zero two

textfile".

38 therefore, we will now remove the slash:

39 */

40 posOfSlash = filepath~pos("-") -- get the position of the slash inside the filepath

41 subStrBeforeSlash = filepath~left(posOfSlash - 1) -- retrieve text from position 0 to just

before the slash

42 subStrAfterSlash = filepath~substr(posOfSlash + 1) -- retrieve the text starting at

position of slash + 1 (so we do not include the slash in this string) until the end of

the variable "filepath"

43 newSpeakableFilePath = subStrBeforeSlash subStrAfterSlash -- merge both variables

44

45 SpeechSynthesizer~Speak(newSpeakableFilePath) -- speak the new file path

46 SpeechSynthesizer~Rate = 1 -- get the rate back up to 1

47

48 DO WHILE mystream~chars > 0 -- read the file until it has reached its end

49 line = myStream~linein -- read an entire line from file

50 DO i = 1 to words(line) -- iterate over words

51 word = word(line,i) -- extract word

52 Console~Write(word || " ") -- display the word

53 SpeechSynthesizer~Speak(word) -- speak the word

54 END

55 END

56

57 ::REQUIRES CLR.CLS -- get ooRexx.NET support

Listing 17: Creating a text-to-speech (TTS) application in ooRexx.NET

Figure 34: This image shows the default output of 15-text.to.speech.rxj

Figure 35: This is another variation of this example. By supplying the application with
an argument, it will override the default text with the argument.

Any programming language with access to the .NET framework has access
to this easy-to-use text-to-speech support provided by the framework. The
needed class is "SpeechSynthesizer" from the namespace "System.Speech.Syn-
thesis". It does not only provide a lot of customization options like the
"rate" of the speaker and the "volume" of the audio output, but also pro-
vides the possibility to select a di�erent speaker (or as Microsoft calls them:
"Voice" [18] ). For the full list of supported languages, see [19, 20, 21].

16 16-GeoLocation.rxj

This application demonstrates how to extract the location data of the user
by using the "System.Device.Location" namespace of the "System.Device"

Assembly. It provides programmers a couple of classes using only one ap-
plication programming interface (API) - independent of the source of the
location data, which Microsoft refer to as the "Location Provider".

1 /* File: 16-GeoLocation.rxj

2 * Description: This application tries to access location data on the device and then

computes the location to an adress.

3 * Shows:

4 * - Setting up a "GeoCoordinateWatcher" based on a high accuracy level

5 * - Accessing permission and status of the GeoCoordinateWatcher class

6 * - Converting ("Resolving") a GeoCoordinate to the exact civic adress

7 */

8

9 CALL clr.addAssembly("System.Device") -- add System.Device Assembly to use all classes

which have a "System.Device.*" namespace

10

11 lf = "0A"x

12 GeoPositionAccuracy = clr.import("System.Device.Location.GeoPositionAccuracy")

13 GeoCoordinateWatcher = .clr~new("System.Device.Location.GeoCoordinateWatcher",

GeoPositionAccuracy~High) -- use the most accurate position possible

14 GeoCoordinateWatcher~clr.dispatch("Start") -- start tracking the location

15

16 jumpcounter = 0 -- set jumpcounter to 0

17 jumpspot:

18 SAY "Accessing Location Provider Information with ooRexx.NET"

19 SAY "-------------------------------------------------------"

20 SAY "Permission:" GeoCoordinateWatcher~Permission~toString

21 SAY "Status:" GeoCoordinateWatcher~Status~toString lf

22 GeoPositionStatus = clr.import("System.Device.Location.GeoPositionStatus") -- get access

to GeoPositionStatus Enumeration

23

24 IF GeoCoordinateWatcher~Status~equals(GeoPositionStatus~Ready) THEN DO

25 GeoPosition = GeoCoordinateWatcher~Position -- everything is set, retrieve location data

26 AdressResolver = .clr~new("System.Device.Location.CivicAddressResolver")

27 GeoCoordinate = GeoPosition~Location

28

29 Adress = AdressResolver~ResolveAdress(GeoCoordinate) -- get the adress based on the

information in GeoCoordinate

30 SAY lf"Your adress:" Adress~AdressLine1~toString

31 SAY "Your building:" Adress~Building~toString

32 SAY "Your postal code:" Adress~PostalCode~toString

33 SAY "Your state or province:" Adress~StateProvince~toString

34 END

35 ELSE IF GeoCoordinateWatcher~Status~equals(GeoPositionStatus~NoData) THEN DO

36 SAY "Sorry, we could not retrieve any location data from your Location Provider."

37 END

38 ELSE IF GeoCoordinateWatcher~Status~equals(GeoPositionStatus~Disabled) THEN DO

39 SAY "Sorry, the Location Provider is disabled. Unfortunately, your device does not

support Location Services."

40 END

41 ELSE IF GeoCoordinateWatcher~Status~equals(GeoPositionStatus~Initializing) THEN DO

42 SAY "Your Location Provider is still initializing."

43 SAY "Trying again ..."

44 CALL SysSleep 3

45 jumpcounter += 1 -- count jumpcounter up

46 IF jumpcounter = 5 THEN EXIT -- if jumpcounter has reached 5 (e.g. after 4 jumps) exit

the application to prevent unnecessary infinity loops

47 CALL jumpspot -- retry after 3 seconds

48 END

49

50 ::REQUIRES CLR.CLS

Listing 18: Retrieving the geographical location of the user with 16-GeoLocation.rxj

Figure 36: This image shows the output of 16-GeoLocation.rxj when access to the
location is not granted there is no GPS sensor installed.

The location may come from a GPS sensor, a wi� spot sharing that in-
formation or other sources. Programmers using the .NET framework do not
have to worry about how they get the location data, they just do. Unfortu-
nately, it is necessary to have some kind of location sensor installed on the
computer, preferably a GPS sensor. The output of status "NoData" in line
21 of Listing 17 indicates that there is no such sensor in the device.
Alternatively, the status "Disabled" would mean that there is a location sen-
sor in the device, but it is deactivated and thus needs some action from
the user. That is why it is very important to check the status of the
"GeoCoordinateWatcher" class instance for all possibilities as shown in this
application. Still, this classes with this namespace is of special value for cre-
ating applications for mobile devices, because they are more likely to have
the needed hardware on-board.
In the case of mobile devices the event "PositionChanged" becomes very in-
teresting. This event gets invoked each time the latitude or longitude of the
user changes. [22]

III Summary, Outlook and Future Research

oorexx.NET is an easy way for interacting with .NET objects and classes by
using Open Object Rexx and BSF4ooRexx. It can be combined with Java
classes because CLR.CLS already requires BSF.CLS - therefore using both
.NET and Java classes will generate no additional loading times or lose ef-
�ciency in any way. Each and every programming language and framework
has its own bene�ts like Java's platform independence or .NET's cryptogra-
phy classes and built-in speech synthesiser. Though using those advantages
is great, combining multiple languages and engines to maximise the outcome
of the application is even better. Once ooRexx.NET gets distributed with
BSF4ooRexx in the fall of 2016 after the beta phase, every ooRexx program-
mer can get his hands on the .NET framework.
As already stated in example "16-GeoLocation.rxj" on page 59 the author
sees a great opportunity in developing applications for mobile devices, like
smartphones or tablets. As of now mobile programming for Windows Phones
o�ers most of the .NET classes and a few additional ones, which would not
make any sense on desktop computers like screen-on time, or access to de-
fault applications like text messaging or calling. It would be very interesting
to see someone develop mobile applications using an ooRexx.NET-related
framework.
All in all the author very much enjoyed using ooRexx.NET for creating the
applications for this bachelor thesis. Microsoft's documentation is great, al-
though the loading times of the MSDN website were rather long, especially
during business hours.

